UPPSALA UNIVERSITET

Tutorial: Introduction to Markov modelling

Mats O Karlsson

Pharmacometrics Research Group
Department of Pharmaceutical Biosciences
Uppsala University
Sweden

UPPSALA UNIVERSITET

Coin toss

Assume a coin has been tossed n times

P_{tails}=1/2 (if perfect coin)

Q: What is the probability of ”tails” at the next toss

P_{n+1}(”tails”) = P_{tails}

Outcome is dependent on P_{tails} only and not dependent on outome of previous tosses

UPPSALA UNIVERSITET

Coin toss

Assume a coin has been tossed n times

P_{tails}=1/2 (perfect coin)

Q: What is the total number of ”tails” (Σ_{n+1}(”tails”)) after the next toss

$P_{n+1}(\Sigma_{n+1}(\text{"tails"})<\Sigma_n(\text{"tails"}))$ = 0

$P_{n+1}(\Sigma_{n+1}(\text{"tails"})=\Sigma_n(\text{"tails"}))$ = P_{tails}

$P_{n+1}(\Sigma_{n+1}(\text{"tails"})=\Sigma_n(\text{"tails"})+1)$ = P_{tails}

$P_{n+1}(\Sigma_{n+1}(\text{"tails"})>\Sigma_n(\text{"tails"})+2)$ = 0

Outcome is dependent on P_{tails} and outomes of previous tosses

Σ_n(”tails”) contains all necessary information about prior history

UPPSALA UNIVERSITET

Definition Markov property

The Markov property, proposed by A.A. Markov (1856-1922), asserts that **the distribution of future outcomes depend only on the current state and not on the whole history**

Modeling of sleep/wake state

- Example: Nighttime observations of awake or sleep every 5th minute in an insomniac patient



UPPSALA UNIVERSITET

Logistic model – NMTRAN

```
$PROB Logistic model for sleep and awake
$DATA data
$INPUT DV
$PRED
   P1 = THETA(1)
   IF(DV.EQ.1) Y=P1
   IF(DV.EQ.0) Y=1-P1
$THETA  (0,.5,1)     ; PROBABILITY OF BEING ASLEEP
$ESTIM LIKE
```

Two-state Markov model



$$P00 = 1 - P01$$

$$P11 = 1 - P10$$

UPPSALA UNIVERSITET

Markov model – NMTRAN code

```
$PROB Transition probabilities between sleep and awake
$DATA data
$INPUT DV PDV                    ;PDV=Previous DV
;PDV = Value of immediately preceding observation
$PRED
  P10 = THETA(1)
  P01 = THETA(2)
  IF(PDV.EQ.0.AND.DV.EQ.1) Y=P01
  IF(PDV.EQ.0.AND.DV.EQ.0) Y=1-P01
  IF(PDV.EQ.1.AND.DV.EQ.0) Y=P10
  IF(PDV.EQ.1.AND.DV.EQ.1) Y=1-P10
$THETA  (0,.1,1)     ; PROB AWAKE GIVEN ASLEEP
$THETA  (0,.1,1)     ; PROB ASLEEP GIVEN AWAKE
$ESTIM LIKE
```

UPPSALA UNIVERSITET

Same model – MLXTRAN code

```
DESCRIPTION:
Categorical data model with Markovian dependence,
   Binomial distribution
INPUT:
parameter = {p01, p11}
OBSERVATION:
Y = {
      type = categorical
      categories = {0,1}
      dependence = Markov
      P(Y=1 | Yp=0) = p01
      P(Y=1 | Yp=1) = p11
}
```

UPPSALA UNIVERSITET

Results

Logistic model

– P1 = 0.43 ± 0.05 OFV 132.7

Markov model

– P01 = 0.11 ± 0.04 OFV 72.4

– P10 = 0.14 ± 0.05

UPPSALA UNIVERSITET

Two-state Markov model



Transition Matrix		
	PDV=0	PDV=1
DV=0	0.89	0.14
DV=1	0.11	0.86

UPPSALA UNIVERSITET

Simulations

Original data



Simulation from logistic model



Simulation from Markov model



UPPSALA UNIVERSITET

Discrete-time Markov model



Continuous-time Markov model



UPPSALA UNIVERSITET

Continuous-time Markov model

```
$PROB Continuous-time two-state Markov model for awake/asleep
$DATA data
$INPUT DV PDV DELTAT
$PRED
  K01 = THETA(1)
  K10 = THETA(2)
  P11    = (EXP(-(K10+K01)*DELTAT)*K10+K01)/(K01+K10)
  P00    = (EXP(-(K10+K01)*DELTAT)*K01+K10)/(K01+K10)
  IF(PDV.EQ.1.AND.DV.EQ.1) Y=P11
  IF(PDV.EQ.1.AND.DV.EQ.0) Y=1-P11
  IF(PDV.EQ.0.AND.DV.EQ.1) Y=P00
  IF(PDV.EQ.0.AND.DV.EQ.0) Y=1-P00
$THETA  (0,1)    ; K ASLEEP GIVEN AWAKE
$THETA  (0,1)    ; K AWAKE GIVEN ASLEEP
$ESTIM LIKE
```

Continuous–time Markov model - differential eqn parametrization

UPPSALA UNIVERSITET

```
$PROB Transition probability between awake and asleep
$DATA data
$INPUT DV TIME AMT  CMT EVID
;          ”Complex” data set to initiate and empty compartments
;            to reset compartment amounts after each observation
$MODEL COMP=PRWAKE COMP=PRSLP
$PK
  K10 = THETA(1)
  K01 = THETA(2)
$DES
DADT(1) = - K01*A(1) + K10*A(2)   ;Represents Probability of 0 - awake
DADT(2) =   K01*A(1) – K10*A(2)   ;Represents Probability of 1 - asleep
$ERROR
   IF(DV.EQ.0) Y  = A(1)
   IF(DV.EQ.1) Y  = A(2)
$THETA  (0,1)     ; K ASLEEP GIVEN AWAKE
$THETA  (0,1)     ; K AWAKE GIVEN ASLEEP
$ESTIM LIKE
```

Data set for differential eqn solution to Markov model

```
$PROB Two-state Markov model
; Entire system updated after each observation
$DATA data
$INPUT DV TIME AMT  CMT EVID
;        .   0   1    1   1    ;awake at start - initialization
;        0   1   0    .   0    ;obs awake at T=1
;        .   1   1    1   4    ;set probabilities to P(awake)=1
;        1   2   0    .   0    ;obs asleep at T=2
;        .   2   1    2   4    ;set probabilities to P(asleep)=1
```

UPPSALA UNIVERSITET

Data set for differential eqn solution to Markov model

```
$PROB Two-state Markov model
;Data set structure with updating compartment by compartment
;May be needed when not entire system is to be updated (e.g. PKPD model)
$DATA data
$INPUT DV TIME AMT  CMT EVID
;        .   0   1    1   1    ;awake at start - initialization
;        0   1   0    .   0    ;awake at T=1
;        .   1   0   -1   2    ; empty compartment 1
;        .   1   0   -2   2    ; empty compartment 2
;        .   1   1    1   1    ; reinitialize comp 1 to 1
;        .   1   0    2   2    ; restart comp 2 with 0
;        1   2   0    .   0    ;asleep at T=2
;        .   2   0   -1   2    ; empty compartment 1
;        .   2   0   -2   2    ; empty compartment 2
;        .   2   1    2   1    ; reinitialize comp 2 to 1
;        .   2   0    1   2    ; restart comp 1 with 0
```

Probabilities following an awake observation at time=0

No additional observations made



$$p00 = \frac{K10}{K10 + K01} + \frac{K01}{K10 + K01} e^{-(K10+K01)t}$$

Probabilities following an awake observation at time=0

Additional observations made at
1 h (awake) and 2 h (asleep)



Introducing a treatment effect

Without treatment



With treatment



Promote falling asleep

K01 ↑

K10 ↔

K01+K10 ↑



Inhibit waking up

K01 ↔

K10 ↓

K01+K10 ↓





Both effects

K01 ↑

K10 ↓

K01+K10 ↔

UPPSALA UNIVERSITET

Clues about mechanism in data

Mechanism	Total sleep time	Average time of sleep episodes	Average time of awake episodes	# of arousals	# of times falling asleep	Total # of State transitions
Promote falling asleep	↑	↔	↓	↑	↑	↑
Inhibit waking up	↑	↑	↔	↓	↓	↓
Both effects K10+K01 ↔	↑	↑	↓	↔	↔	↔

Repeated time to event analysis
an alternative to Markov model

- Assume transitions as events
- Assume no unobserved transitions
- Separate (constant) hazards for waking up and falling asleep



X – event during awake period (falling asleep) I – event during sleeping period (arousal)
0 – censoring during awake period

RTTE model – NMTRAN

```
$PROB Repeated Time To Event data - constant hazard
$INPUT ID DV TYPE DVID
$DATA data
$PRED
 IF(TYPE.EQ.0) HAZ= THETA(1)  ;hazard for waking up
 IF(TYPE.EQ.1) HAZ= THETA(2)  ;hazard for falling asleep
 CUMHAZ    = HAZ*DV           ;cumulative hazard
 SUR       = EXP(-CUMHAZ)     ;survival probability
 IF(DVID.EQ.1) Y  = HAZ*SUR   ;event
 IF(DVID.EQ.2) Y  = SUR       ;censoring
$THETA (0,.146)   ; HAZ_WK (/5min)
$THETA (0,.109)   ; HAZ_SL (/5min)
$ESTIM LIKE
```

UPPSALA UNIVERSITET

Markov models @ PAGE



UPPSALA UNIVERSITET

Pharmacometric Markov models

Bergstrand	GI movement	Non-ordered categorical	Compartmental chain	2009	CPT
Bizzotto	Sleep	Non-ordered categorical	1st order + stagetime	2008	PAGE
Girard	Adherence	3-state categorical	Discrete-time	1998	Stat Med
Henin	Hand-and-foot syndrome	Ordered categorical	1st order	2008	CPT
Ito	Dizziness	Ordered categorical	1st order	2008	CPT
Karlsson	Sleep	Non-ordered categorical	1st order + stagetime	2000	CPT
Karlsson	Sedation	Ordered categorical	1st order +stagetime	2002	Measurement and kinetics of in vivo drug effects
Kjellsson	Sleep	Non-ordered categorical	1st order + stagetime	2006	PAGE
Lacroix	ACR20	Binary + dropout	1st order	2009	CPT
Maas	Migraine	Ordered categorical	Hidden Markov	2006	Cephalalgia
Plan	None	Count	1st order	2009	JPKPD
Snoeck	Seizures	Count	1st order	2007	PAGE
Troconiz	Seizures	Count	1st order	2007	PAGE
Zandvliet	Follicules	Multinomial count	Compartmental chain	2008	PAGE
Zingmark	Side-effect	Ordered categorical	1st order	2005	JPKPD

Markov model for responder, non-responder and dropout

Ex, ACR20 score in Rheumatoid Arthritis



Absorbing state – cannot leave the state

Absorbing chain – absorbing state can be reached from all other states

Lacroix et al., Clin Pharmacol Ther, 2009

NMTRAN code
Responder, non-responder and dropout

```
$PRED
;----transition from being a responder to non-responder---
LGT01=LOG(THETA(1)/(1-THETA(1))) + ETA(1)
P01=EXP(LGT01)/(1+EXP(LGT01))

;----transition from responder to dropout---
LGT21=LOG(THETA(2)/(1-THETA(2)))
P21=EXP(LGT21)/(1+EXP(LGT21))

;-- transitions from being a non-responder to a responder---
LGT10=LOG(THETA(3)/(1-THETA(3))) + ETA(2)
P10=EXP(LGT10)/(1+EXP(LGT10))

;---transition from non-responder to dropout---
LGT20=LOG(THETA(2)/(1-THETA(2)))
P20=EXP(LGT20)/(1+EXP(LGT20))
```

```
;-------- transition's probabilities ----------
IF (PREV.EQ.1.AND.DV.EQ.0) Y=P01
IF (PREV.EQ.1.AND.DV.EQ.2) Y=P21*(1-P01)
IF (PREV.EQ.1.AND.DV.EQ.1) Y=1-P01-P21*(1-P01)
IF (PREV.EQ.0.AND.DV.EQ.1) Y=P10
IF (PREV.EQ.0.AND.DV.EQ.2) Y=P20*(1-P10)
IF (PREV.EQ.0.AND.DV.EQ.0) Y=1-P10-P20*(1-P10)

$EST  METH=1 LAPLACE LIKE
```

Diagnostics – ACR20 model - proportion responders

UPPSALA UNIVERSITET



Lacroix et al., Clin Pharmacol Ther, 2009

Diagnostics – ACR20 model - transitions



Spontaneous reporting of a side-effect



Zingmark et al., JPKPD 2005

Alternative assumptions regarding nature of transitions





For RTT(C)E implementation see Plan et al., CP&T 2010

UPPSALA UNIVERSITET

Parameterization of the model

$$[l_{S\geq1}|pre=0]=b_1+D$$
$$[l_{S=2}|pre=0]=b_1+b_2+D$$
$$[l_{S\geq1}|pre=1]=b_3+D$$
$$[l_{S=2}|pre=1]=b_3+b_4+D$$
$$[l_{S\geq1}|pre=2]=b_5+D$$
$$[l_{S=2}|pre=2]=b_5+b_6+D$$

$$PC_x=\frac{e^{l_x}}{1+e^{l_x}}$$

$p_{S=0}$ = $1-PC_{S\geq1}$

$p_{S=1}$ = $PC_{S\geq1}-PC_{S=2}$

$p_{S=2}$ = $PC_{S=2}$

$$D=\frac{E_{\max}\Big|_{pre=0,1,2}\cdot e^{\eta_i}\cdot Ce}{EC_{50}\cdot(1+C_{tol}/TC_{50})+Ce}$$

Spontaneous reporting of a side-effect - model simulations

UPPSALA UNIVERSITET

	Proportional odds model			Markov model			
Number of transitions	**1 min**	**3 min**	**6 min**	**1 min**	**3 min**	**6 min**	**Observed data**
0-2	289 (249; 335)	117 (102; 133)	72 (62; 82)	11 (9; 14)	11 (8; 15)	11 (8; 15)	11
1-2	178 (133; 223)	79 (64; 95)	48 (39; 59)	25 (21; 29)	25 (20; 29)	25 (21; 29)	23

The results shown are the average and (10^{th} and 90^{th} percentiles) from 100 simulated datasets

The performance of the proportional odds model, but not the Markov model, is dependent on the choice of observation frequency

UPPSALA UNIVERSITET

Sleep scoring – non-ordered categorical data



(Karlsson et al., CPT 2000)

Choice of transitions to model

- To reduce the number of transitions to model, three criteria were defined to identify the transitions of interest representing:
  - (i) >1% of all observations in a stage,
  - (ii) >10% of all transitions from a stage
  - (iii) >10% of all transitions to a stage
  - A transition was modeled, if at least one of these criteria was fulfilled

(Kjellsson et al., PAGE 2007)

UPPSALA
UNIVERSITET

Overall sleep pattern



(Karlsson et al., CPT 2000)

Transition dependences



UPPSALA UNIVERSITET

What previous info to condition on?

- First-order Markov models often sufficient
- Sometimes biological processes dictate use of higher-order Markov elements
  - 2nd (3rd etc) – order elements
  - Time since entering present stage (stagetime)
  - Prior stage(s)

UPPSALA UNIVERSITET

How to start a Markov chain?

- Sometimes obvious –
  - Sleep-wake data that starts at bed-time
  - Responder status for ACR20
- Sometimes screening data provide information about initial state
- Transform first observation(s) to a covariate for starting the chain

Sedation scores following stroke



Karlsson et al., in Measurement and Kinetics of in vivo drug effects, 2002

Alternative assumptions regarding nature of transitions



Many transitions to model
No assumption about intermediate transition states



Fewer transitions to model
Assumption about intermediate transition states

Individual time courses of sedation

UPPSALA UNIVERSITET

Observations



Markov model simulations



Ordered categorical model simulations



Model for GI transit



Bergstrand et al., CPT 2009

UPPSALA UNIVERSITET

GI position





Count data - daily seizure scores in epileptic patients





Markov model; Troconiz et al., JPKPD 36:461-77 (2009)
Hidden Markov model; Delattre et al., JPKPD 39:263-71 (2012)

Consequences of ignoring Markov properties - estimation

- Information content in data overestimated
  - SEs underestimated
  - Hypothesis tests inappropriate
- Interindividual variability overestimated
- Potential structural model misspecification
- No info on time-course of dependence

UPPSALA UNIVERSITET

Consequences of ignoring Markov properties – simulation & design

Duration of state periods too low

Inflated number of transitions

Inflated number of extreme value occurrencies

- E.g. distribution of maximum severity score in population

Individualisation strategy suboptimal

Value of more frequent observations overrated

Positioning of observation times

Optimal design results inappropriate*

*Optimal design w/ Markov models; Nyberg & Hooker PAGE 2011

UPPSALA UNIVERSITET

Markov model or not?

Start with Markov

- Frequent observations
- Many consequtive same-state observations
- Many levels of response
- Non-ordered categorical data

Start without Markov and diagnose

- Check number of transitions
- Check average duration of same-state periods

UPPSALA UNIVERSITET

Thank you!